

RBC model

Endogenous variables

C_t	consumption
K_t	capital at the end of period
L_t	labor services
w_t	wage rate
r_t	rate of return on capital
A_t	total factor productivity

Exogenous variables

e_t	TFP innovation
-------	----------------

Parameters calibration

alpha	capital share	0.33
delta	depreciation rate	0.025
beta	discount factor	0.9975
lambda	TFP autocorrelation	0.97
gamma	curvature of disutility of labor	0
g	growth rate	0.015

Dynare implementation

```
var C K L w r A;
```

```
varexo e;
```

```
parameters beta delta gamma alpha lambda g;
```

```
alpha = 0.33;
```

```
delta = 0.025;
```

```
beta = 0.9975;
```

```
lambda = 0.97;
```

```
gamma = 0;
```

```
g = 0.015;
```

Dynare implementation (continued)

```
model;  
1/C=beta*(1/(C(+1)*(1+g)))*(r(+1)+1-delta);  
L^gamma = w/C;  
r = alpha*A*(K(-1)/(1+g))^(alpha-1)*L^(1-alpha);  
w = (1-alpha)*A*(K(-1)/(1+g))^alpha*L^(-alpha);  
K+C = A*(K(-1)/(1+g))^alpha*L^(1-alpha) +  
      (1-delta)*(K(-1)/(1+g));  
log(A) = lambda*log(A(-1))+e;  
end;
```

Dynare implementation (continued)

```
steady_state_model;  
A = 1;  
r = (1+g)/beta+delta-1;  
KL = (1+g)*(r/(alpha*A))^(1/(alpha-1));  
w = (1-alpha)*A*(KL/(1+g))^alpha;  
C = w;  
L = C/(A*(KL/(1+g))^alpha  
      + (1-delta)*KL/(1+g) - KL);  
K = KL*L;  
end;  
  
steady;
```

Dynare implementation (continued)

```
shocks;  
var e; stderr 0.01;  
end;  
  
check;  
  
stoch_simul(order=1);
```

Decision and transition functions

Dynare output:

POLICY AND TRANSITION FUNCTIONS

	C	K	L	w	r	A
Constant	1.837697	20.976677	0.971392	1.837697	0.042544	1.000000
K(-1)	0.042056	0.921754	-0.021057	0.042056	-0.001977	0
A(-1)	1.083571	3.558433	1.119646	1.083571	0.074122	0.970000
e	1.117083	3.668488	1.154275	1.117083	0.076415	1.000000

$$C_t = 1.838 + 0.042 (K_{t-1} - \bar{K}) + 1.084 (A_{t-1} - \bar{A}) + 1.118e_t$$

Solving numerically for the steady state

```
initval;  
A = 1;  
r = 0.04;  
w = 1.8;  
C = 1.8;  
L = 1;  
K = 20;  
end;  
  
steady;
```

See rbc1a.mod